@echo off
chcp 65001
TITLE "一键运行"

@set ori_path=%path%
set path=%cd%\bin;%path%

echo.
echo ========================================
echo 是否执行预操作: 开启日志落盘 + 禁用selinux + 清理日志
echo.
echo 输入1，是
echo.
echo 输入2，否
echo.
echo 输入3，已有日志，只做转换
echo ========================================
echo.
set /p pre_init=请输入选择：

if %pre_init% equ 1 goto init
if %pre_init% equ 2 goto no_init
if %pre_init% equ 3 goto trans

:init

echo ========================================
echo 禁用selinux
hdc shell mount -o rw,remount /
@if errorlevel 1 goto error
hdc shell "sed -i 's/enforcing/permissive/g' /system/etc/selinux/config"
@if errorlevel 1 goto error
hdc shell "cat /system/etc/selinux/config |grep SELINUX="

echo ========================================
echo 删除存量日志
hdc shell rm -rf /data/log/hilog/*

echo ========================================
echo 落盘日志
hdc shell hilog -w start
hdc shell hilog -w start -t kmsg

echo ========================================
echo 重启中，重启后按任意键继续
hdc shell reboot
pause

:no_init

echo ========================================
echo 请执行用例，执行后按任意键继续
pause

echo ========================================
echo 获取日志中

if exist log (
    rd /s/q log
)
if exist format_out (
    rd /s/q format_out
)
if exist sort_out (
    rd /s/q sort_out
)
if exist out (
    rd /s/q out
)

@set Dir=log
md %Dir%

for /f "tokens=1" %%a in ('hdc shell ls /data/log/hilog') do (
	hdc file recv /data/log/hilog/%%a .\%Dir%\
	gzip -d .\%Dir%\%%a
)
@if errorlevel 1 goto error

:trans
echo ========================================
echo 日志转换中
python .\trans_avc_rule.py log
@if errorlevel 1 goto error


echo.
echo.
echo ========================================
echo 日志转换完成，输出结果在format_out下
echo.

:success
echo ==============success===================
pause
goto end


:error
echo ==============fail======================
pause

:end
